Rules for integrands of the form $(a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Sec}[e + f x])^n (A + B \operatorname{Sec}[e + f x])$

1. $\int (a + b \operatorname{Sec}[e + f x]) (d \operatorname{Sec}[e + f x])^n (A + B \operatorname{Sec}[e + f x]) \, dx$ when $A b - a B \neq 0$

1: $\int (a + b \operatorname{Sec}[e + f x]) (d \operatorname{Sec}[e + f x])^n (A + B \operatorname{Sec}[e + f x]) \, dx$ when $A b - a B \neq 0 \land n \leq -1$

- Derivation: Nondegenerate secant recurrence 1a with $A \to a A,\ B \to A b + a B,\ C \to b B,\ m \to 0,\ p \to 0$
- Rule: If $A b - a B \neq 0 \land n \leq -1$, then

$$\int (a + b \operatorname{Sec}[e + f x]) (d \operatorname{Sec}[e + f x])^n (A + B \operatorname{Sec}[e + f x]) \, dx \longrightarrow$$
$$-\frac{A a \operatorname{Tan}[e + f x] (d \operatorname{Sec}[e + f x])^n}{f n} + \frac{1}{d n} \int (d \operatorname{Sec}[e + f x])^{n+1} (n (B a + A b) + (B b n + A a (n + 1)) \operatorname{Sec}[e + f x]) \, dx$$

- Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  A*a*Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*n) +
  1/(d*n)*Int[(d*Csc[e+f*x])^(n+1)*Simp[n*(B*a+A*b)+(B*b*n+A*a*(n+1))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && LeQ[n,-1]
```

2: $\int (a + b \operatorname{Sec}[e + f x]) (d \operatorname{Sec}[e + f x])^n (A + B \operatorname{Sec}[e + f x]) \, dx$ when $A b - a B \neq 0 \land n \nleq -1$

- Derivation: Nondegenerate secant recurrence 1b with $A \to a A,\ B \to A b + a B,\ C \to b B,\ m \to 0,\ p \to 0$
- Rule: If $A b - a B \neq 0 \land n \nleq -1$, then

$$\int (a + b \operatorname{Sec}[e + f x]) (d \operatorname{Sec}[e + f x])^n (A + B \operatorname{Sec}[e + f x]) \, dx \longrightarrow$$
$$\frac{b B \operatorname{Tan}[e + f x] (d \operatorname{Sec}[e + f x])^n}{f (n + 1)} + \frac{1}{n + 1} \int (d \operatorname{Sec}[e + f x])^n (A a (n + 1) + B b n + (A b + B a) (n + 1) \operatorname{Sec}[e + f x]) \, dx$$

- Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])*(d_.*csc[e_.+f_.*x_])^n_.*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -b*B*Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*(n+1)) +
  1/(n+1)*Int[(d*Csc[e+f*x])^n*Simp[A*a*(n+1)+B*b*n+(A*b+B*a)*(n+1)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && Not[LeQ[n,-1]]
```

2. $\int \operatorname{Sec}[e + f x] (a + b \operatorname{Sec}[e + f x])^m (A + B \operatorname{Sec}[e + f x]) \, dx$ when $A b - a B \neq 0$

1: $\int \frac{\text{Sec}[e+f x]\,(A+B\,\text{Sec}[e+f x])}{a+b\,\text{Sec}[e+f x]}\,dx$ when $A b-a B\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B z}{a+b z}==\frac{B}{b}+\frac{A b-a B}{b\,(a+b z)}$

Rule: If $A b-a B\neq 0$**, then**

$$\int \frac{\text{Sec}[e+f x]\,(A+B\,\text{Sec}[e+f x])}{a+b\,\text{Sec}[e+f x]}\,dx\ \longrightarrow\ \frac{B}{b}\int \text{Sec}[e+f x]\,dx+\frac{A b-a B}{b}\int \frac{\text{Sec}[e+f x]}{a+b\,\text{Sec}[e+f x]}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(A_+B_.*csc[e_.+f_.*x_])/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  B/b*Int[Csc[e+f*x],x] + (A*b-a*B)/b*Int[Csc[e+f*x]/(a+b*Csc[e+f*x]),x] /;
FreeQ[{a,b,e,f,A,B},x] && NeQ[A*b-a*B,0]
```

2. $\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(A+B\,\text{Sec}[e+f x])\,dx$ when $A b-a B\neq 0\ \land\ a^2-b^2==0$

1: $\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(A+B\,\text{Sec}[e+f x])\,dx$ when $A b-a B\neq 0\ \land\ a^2-b^2==0\ \land\ a B m+A b\,(m+1)==0$

Derivation: Singly degenerate secant recurrence 2a with $A\to -\frac{a B m}{b\,(m+1)},\ n\to 0,\ p\to 0$

Derivation: Singly degenerate secant recurrence 2c with $A\to -\frac{a B m}{b\,(m+1)},\ n\to 0,\ p\to 0$

Note: If $a^2-b^2==0\ \land\ a B m+A b\,(m+1)==0$**, then** $m+1\neq 0$**.**

Rule: If $A b-a B\neq 0\ \land\ a^2-b^2==0\ \land\ a B m+A b\,(m+1)==0$**, then**

$$\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(A+B\,\text{Sec}[e+f x])\,dx\ \longrightarrow\ \frac{B\,\text{Tan}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m}{f\,(m+1)}$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -B*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(f*(m+1)) /;
FreeQ[{a,b,A,B,e,f,m},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && EqQ[a*B*m+A*b*(m+1),0]
```

2. $\int \operatorname{Sec}[e+f x]\,(a+b \operatorname{Sec}[e+f x])^m\,(A+B \operatorname{Sec}[e+f x])\,dx$ when $A b-a B\neq 0 \land a^2-b^2==0 \land a B m+A b (m+1)\neq 0$

1: $\int \operatorname{Sec}[e+f x]\,(a+b \operatorname{Sec}[e+f x])^m\,(A+B \operatorname{Sec}[e+f x])\,dx$ when $A b-a B\neq 0 \bigwedge a^2-b^2==0 \bigwedge a B m+A b (m+1)\neq 0 \bigwedge m<-\frac{1}{2}$

Derivation: Singly degenerate secant recurrence 2a with $n\to 0,\ p\to 0$

Rule: If $A b-a B\neq 0 \bigwedge a^2-b^2==0 \bigwedge a B m+A b (m+1)\neq 0 \bigwedge m<-\frac{1}{2}$, then

$$\int \operatorname{Sec}[e+f x]\,(a+b \operatorname{Sec}[e+f x])^m\,(A+B \operatorname{Sec}[e+f x])\,dx \longrightarrow$$
$$-\frac{(A b-a B) \operatorname{Tan}[e+f x]\,(a+b \operatorname{Sec}[e+f x])^m}{a f (2 m+1)}+\frac{a B m+A b (m+1)}{a b (2 m+1)}\int \operatorname{Sec}[e+f x]\,(a+b \operatorname{Sec}[e+f x])^{m+1}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  (A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(a*f*(2*m+1)) +
  (a*B*m+A*b*(m+1))/(a*b*(2*m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1),x] /;
FreeQ[{a,b,A,B,e,f},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && NeQ[a*B*m+A*b*(m+1),0] && LtQ[m,-1/2]
```

2: $\int \operatorname{Sec}[e+f x]\,(a+b \operatorname{Sec}[e+f x])^m\,(A+B \operatorname{Sec}[e+f x])\,dx$ when $A b-a B\neq 0 \bigwedge a^2-b^2==0 \bigwedge a B m+A b (m+1)\neq 0 \bigwedge m\nless-\frac{1}{2}$

Derivation: Singly degenerate secant recurrence 2c with $n\to 0,\ p\to 0$

Rule: If $A b-a B\neq 0 \bigwedge a^2-b^2==0 \bigwedge a B m+A b (m+1)\neq 0 \bigwedge m\nless-\frac{1}{2}$, then

$$\int \operatorname{Sec}[e+f x]\,(a+b \operatorname{Sec}[e+f x])^m\,(A+B \operatorname{Sec}[e+f x])\,dx \longrightarrow$$
$$\frac{B \operatorname{Tan}[e+f x]\,(a+b \operatorname{Sec}[e+f x])^m}{f (m+1)}+\frac{a B m+A b (m+1)}{b (m+1)}\int \operatorname{Sec}[e+f x]\,(a+b \operatorname{Sec}[e+f x])^m\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -B*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(f*(m+1)) +
  (a*B*m+A*b*(m+1))/(b*(m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m,x] /;
FreeQ[{a,b,A,B,e,f,m},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && NeQ[a*B*m+A*b*(m+1),0] && Not[LtQ[m,-1/2]]
```

3. $\int Sec[e+f x] (a+b Sec[e+f x])^m (A+B Sec[e+f x]) \, dx$ when $A b-a B \neq 0 \land a^2-b^2 \neq 0$

1: $\int Sec[e+f x] (a+b Sec[e+f x])^m (A+B Sec[e+f x]) \, dx$ when $A b-a B \neq 0 \land a^2-b^2 \neq 0 \land m>0$

Reference: G&R 2.551.1 inverted

Derivation: Nondegenerate secant recurrence 1b with $A \to a A,\ B \to A b+a B,\ C \to b B,\ m \to 0,\ n \to n-1,\ p \to 0$

Rule: If $A b-a B \neq 0 \land a^2-b^2 \neq 0 \land m>0$**, then**

$$\int Sec[e+f x] (a+b Sec[e+f x])^m (A+B Sec[e+f x]) \, dx \longrightarrow$$

$$\frac{B\,Tan[e+f x] (a+b Sec[e+f x])^m}{f (m+1)}+\frac{1}{m+1} \int Sec[e+f x] (a+b Sec[e+f x])^{m-1} (b B m+a c (m+1)+(a B m+A b (m+1)) Sec[e+f x]) \, dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -B*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(f*(m+1)) +
  1/(m+1)*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m-1)*Simp[b*B*m+a*A*(m+1)+(a*B*m+A*b*(m+1))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,A,B,e,f},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && GtQ[m,0]
```

2: $\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1$

Reference: G&R 2.551.1

Derivation: Nondegenerate secant recurrence 1a with $C\to 0,\ n\to 0,\ p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1$**, then**

$$\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x])\,dx\longrightarrow$$

$$\frac{(A\,b-a\,B)\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m+1}}{f\,(m+1)\,\left(a^2-b^2\right)}+$$

$$\frac{1}{(m+1)\,\left(a^2-b^2\right)}\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m+1}\,((a\,A-b\,B)\,(m+1)-(A\,b-a\,B)\,(m+2)\,\text{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(f*(m+1)*(a^2-b^2)) +
  1/((m+1)*(a^2-b^2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*Simp[(a*A-b*B)*(m+1)-(A*b-a*B)*(m+2)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,A,B,e,f},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[m,-1]
```

3. $\int \frac{\text{Sec}[e+f\,x]\,(A+B\,\text{Sec}[e+f\,x])}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0$

1: $\int \frac{\text{Sec}[e+f\,x]\,(A+B\,\text{Sec}[e+f\,x])}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0\,\wedge\,A^2-B^2==0$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x\left(\frac{1}{\text{Tan}[e+f\,x]}\sqrt{\frac{b\,(1-\text{Sec}[e+f\,x])}{a+b}}\,\sqrt{-\frac{b\,(1+\text{Sec}[e+f\,x])}{a-b}}\right)==0$

- **Basis:** $\text{Sec}[e+f\,x]\,\text{Tan}[e+f\,x]\,F[\text{Sec}[e+f\,x]]==\frac{1}{f}\,\text{Subst}[F[x],\,x,\,\text{Sec}[e+f\,x]]\,\partial_x\text{Sec}[e+f\,x]$

Rule: If $a^2-b^2\neq 0$**, then**

$$\int \frac{\text{Sec}[e+f\,x]\,(A+B\,\text{Sec}[e+f\,x])}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx\longrightarrow\frac{A\,b-a\,B}{b\,\text{Tan}[e+f\,x]}$$

$$\sqrt{\frac{b\,(1-\text{Sec}[e+f\,x])}{a+b}}\;\sqrt{-\frac{b\,(1+\text{Sec}[e+f\,x])}{a-b}}\int\frac{\text{Sec}[e+f\,x]\;\text{Tan}[e+f\,x]\;\sqrt{-\frac{b\,B}{a\,A-b\,B}-\frac{A\,b\,\text{Sec}[e+f\,x]}{a\,A-b\,B}}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\;\sqrt{\frac{b\,B}{a\,A+b\,B}-\frac{A\,b\,\text{Sec}[e+f\,x]}{a\,A+b\,B}}}\,dx$$

$$\rightarrow\;\frac{A\,b-a\,B}{b\,f\,\text{Tan}[e+f\,x]}\sqrt{\frac{b\,(1-\text{Sec}[e+f\,x])}{a+b}}\;\sqrt{-\frac{b\,(1+\text{Sec}[e+f\,x])}{a-b}}\;\text{Subst}\Big[\int\frac{\sqrt{-\frac{b\,B}{a\,A-b\,B}-\frac{A\,b\,x}{a\,A-b\,B}}}{\sqrt{a+b\,x}\;\sqrt{\frac{b\,B}{a\,A+b\,B}-\frac{A\,b\,x}{a\,A+b\,B}}}\,dx,\;x,\;\text{Sec}[e+f\,x]\Big]$$

$$\rightarrow\;\frac{2\,(A\,b-a\,B)\sqrt{a+\frac{b\,B}{A}}\;\sqrt{\frac{b\,(1-\text{Sec}[e+f\,x])}{a+b}}\;\sqrt{-\frac{b\,(1+\text{Sec}[e+f\,x])}{a-b}}}{b^2\,f\,\text{Tan}[e+f\,x]}\;\text{EllipticE}\Big[\text{ArcSin}\Big[\frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{a+\frac{b\,B}{A}}}\Big],\;\frac{a\,A+b\,B}{a\,A-b\,B}\Big]$$

Program code:

```
Int[csc[e_.+f_.*x_]*(A_+B_.*csc[e_.+f_.*x_])/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -2*(A*b-a*B)*Rt[a+b*B/A,2]*Sqrt[b*(1-Csc[e+f*x])/(a+b)]*Sqrt[-b*(1+Csc[e+f*x])/(a-b)]/(b^2*f*Cot[e+f*x])*
    EllipticE[ArcSin[Sqrt[a+b*Csc[e+f*x]]/Rt[a+b*B/A,2]],(a*A+b*B)/(a*A-b*B)] /;
FreeQ[{a,b,e,f,A,B},x] && NeQ[a^2-b^2,0] && EqQ[A^2-B^2,0]
```

2: $\int \frac{\text{Sec}[e+f x] (A+B \text{Sec}[e+f x])}{\sqrt{a+b \text{Sec}[e+f x]}} dx$ when $a^2-b^2 \neq 0 \land A^2-B^2 \neq 0$

Derivation: Algebraic expansion

Basis: $A+B z == A-B+B (1+z)$

Rule: If $a^2-b^2 \neq 0 \land A^2-B^2 \neq 0$**, then**

$$\int \frac{\text{Sec}[e+f x] (A+B \text{Sec}[e+f x])}{\sqrt{a+b \text{Sec}[e+f x]}} dx \to (A-B) \int \frac{\text{Sec}[e+f x]}{\sqrt{a+b \text{Sec}[e+f x]}} dx + B \int \frac{\text{Sec}[e+f x] (1+\text{Sec}[e+f x])}{\sqrt{a+b \text{Sec}[e+f x]}} dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(A_+B_.*csc[e_.+f_.*x_])/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  (A-B)*Int[Csc[e+f*x]/Sqrt[a+b*Csc[e+f*x]],x] +
  B*Int[Csc[e+f*x]*(1+Csc[e+f*x])/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,e,f,A,B},x] && NeQ[a^2-b^2,0] && NeQ[A^2-B^2,0]
```

4: $\int \text{Sec}[e+f x] (a+b \text{Sec}[e+f x])^m (A+B \text{Sec}[e+f x]) dx$ when $A b-a B \neq 0 \land a^2-b^2 \neq 0 \land A^2-B^2 == 0 \land 2 m \notin \mathbb{Z}$

Derivation: Integration by substitution

Rule: If $A b-a B \neq 0 \land a^2-b^2 \neq 0 \land A^2-B^2 == 0 \land 2 m \notin \mathbb{Z}$**, then**

$$\int \text{Sec}[e+f x] (a+b \text{Sec}[e+f x])^m (A+B \text{Sec}[e+f x]) dx \to$$

$$-\frac{2 \sqrt{2} A (a+b \text{Sec}[e+f x])^m (A-B \text{Sec}[e+f x]) \sqrt{\frac{A+B \text{Sec}[e+f x]}{A}}}{B f \text{Tan}[e+f x] \left(\frac{A (a+b \text{Sec}[e+f x])}{a A+b B}\right)^m} \text{AppellF1}\left[\frac{1}{2}, -\frac{1}{2}, -m, \frac{3}{2}, \frac{A-B \text{Sec}[e+f x]}{2 A}, \frac{b (A-B \text{Sec}[e+f x])}{A b+a B}\right]$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  2*Sqrt[2]*A*(a+b*Csc[e+f*x])^m*(A-B*Csc[e+f*x])*Sqrt[(A+B*Csc[e+f*x])/A]/(B*f*Cot[e+f*x]*(A*(a+b*Csc[e+f*x])/(a*A+b*B))^m)*
    AppellF1[1/2,-(1/2),-m,3/2,(A-B*Csc[e+f*x])/(2*A),(b*(A-B*Csc[e+f*x]))/(A*b+a*B)] /;
FreeQ[{a,b,A,B,e,f},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && EqQ[A^2-B^2,0] && Not[IntegerQ[2*m]]
```

5: $\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(A+B\,\text{Sec}[e+f x])\,dx$ when $A b-a B\neq 0\,\wedge\,a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $A+B z = \frac{A b-a B}{b}+\frac{B}{b}(a+b z)$

Rule: If $A b-a B\neq 0\,\wedge\,a^2-b^2\neq 0$, **then**

$$\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(A+B\,\text{Sec}[e+f x])\,dx \longrightarrow \frac{A b-a B}{b}\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,dx+\frac{B}{b}\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^{m+1}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  (A*b-a*B)/b*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m,x] + B/b*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1),x] /;
FreeQ[{a,b,A,B,e,f,m},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0]
```

3. $\int \text{Sec}[e+f x]^2\,(a+b\,\text{Sec}[e+f x])^m\,(A+B\,\text{Sec}[e+f x])\,dx$ when $A b-a B\neq 0$

1: $\int \text{Sec}[e+f x]^2\,(a+b\,\text{Sec}[e+f x])^m\,(A+B\,\text{Sec}[e+f x])\,dx$ when $A b-a B\neq 0\,\wedge\,a^2-b^2=0\,\wedge\,m<-\frac{1}{2}$

Derivation: ???

Rule: If $A b-a B\neq 0\,\wedge\,a^2-b^2=0\,\wedge\,m<-\frac{1}{2}$, **then**

$$\int \text{Sec}[e+f x]^2\,(a+b\,\text{Sec}[e+f x])^m\,(A+B\,\text{Sec}[e+f x])\,dx \longrightarrow$$

$$\frac{(A b-a B)\,\text{Tan}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m}{b f\,(2 m+1)}+$$

$$\frac{1}{b^2\,(2 m+1)}\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^{m+1}\,(m\,(A b-a B)+b B\,(2 m+1)\,\text{Sec}[e+f x])\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(b*f*(2*m+1)) +
  1/(b^2*(2*m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*Simp[A*b*m-a*B*m+b*B*(2*m+1)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && LtQ[m,-1/2]
```

2: $\int \mathrm{Sec}[e+f x]^2 (a+b\,\mathrm{Sec}[e+f x])^m (A+B\,\mathrm{Sec}[e+f x])\, dx$ when $A b-a B\neq 0 \land a^2-b^2\neq 0 \land m<-1$

Derivation: Nondegenerate secant recurrence 1a with $A\to a A,\ B\to A b+a B,\ C\to b B,\ m\to 0,\ p\to 0$

Rule: If $A b-a B\neq 0 \land a^2-b^2\neq 0 \land m<-1$, then

$$\int \mathrm{Sec}[e+f x]^2 (a+b\,\mathrm{Sec}[e+f x])^m (A+B\,\mathrm{Sec}[e+f x])\, dx \longrightarrow$$

$$-\frac{a (A b-a B) \mathrm{Tan}[e+f x] (a+b\,\mathrm{Sec}[e+f x])^{m+1}}{b f (m+1) \left(a^2-b^2\right)}-$$

$$\frac{1}{b (m+1) \left(a^2-b^2\right)}\int \mathrm{Sec}[e+f x] (a+b\,\mathrm{Sec}[e+f x])^{m+1} \left(b (A b-a B) (m+1)-\left(a A b (m+2)-B \left(a^2+b^2 (m+1)\right)\right) \mathrm{Sec}[e+f x]\right) dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  a*(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b*f*(m+1)*(a^2-b^2)) -
  1/(b*(m+1)*(a^2-b^2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*
    Simp[b*(A*b-a*B)*(m+1)-(a*A*b*(m+2)-B*(a^2+b^2*(m+1)))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[m,-1]
```

3: $\int \mathrm{Sec}[e+f x]^2 (a+b\,\mathrm{Sec}[e+f x])^m (A+B\,\mathrm{Sec}[e+f x])\, dx$ when $A b-a B\neq 0 \land m \nless -1$

Derivation: Nondegenerate secant recurrence 1b with $A\to a A,\ B\to A b+a B,\ C\to b B,\ m\to 0,\ p\to 0$

Rule: If $A b-a B\neq 0 \land m \nless -1$, then

$$\int \mathrm{Sec}[e+f x]^2 (a+b\,\mathrm{Sec}[e+f x])^m (A+B\,\mathrm{Sec}[e+f x])\, dx \longrightarrow$$

$$\frac{B\,\mathrm{Tan}[e+f x] (a+b\,\mathrm{Sec}[e+f x])^{m+1}}{b f (m+2)}+\frac{1}{b (m+2)}\int \mathrm{Sec}[e+f x] (a+b\,\mathrm{Sec}[e+f x])^m (b B (m+1)+(A b (m+2)-a B) \mathrm{Sec}[e+f x])\, dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -B*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b*f*(m+2)) +
  1/(b*(m+2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m*Simp[b*B*(m+1)+(A*b*(m+2)-a*B)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,m},x] && NeQ[A*b-a*B,0] && Not[LtQ[m,-1]]
```

4. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2==0$

1. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,m+n+1==0$

1: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,m+n+1==0\,\wedge\,a\,A\,m-b\,B\,n==0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,m+n+1==0\,\wedge\,a\,A\,m-b\,B\,n==0$**, then**

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\,\longrightarrow\,-\frac{A\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n}{f\,n}$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*n) /;
FreeQ[{a,b,d,e,f,A,B,m,n},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && EqQ[m+n+1,0] && EqQ[a*A*m-b*B*n,0]
```

2. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,m+n+1==0\,\wedge\,a\,A\,m-b\,B\,n\neq 0$

1: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,m+n+1==0\,\wedge\,m\leq -1$

Derivation: Singly degenerate secant recurrence 2b with $m\to -n-2,\ p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,m+n+1==0\,\wedge\,m\leq -1$**, then**

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\,\longrightarrow$$
$$\frac{(A\,b-a\,B)\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n}{b\,f\,(2\,m+1)}+\frac{(a\,A\,m+b\,B\,(m+1))}{a^2\,(2\,m+1)}\int (a+b\,\text{Sec}[e+f\,x])^{m+1}\,(d\,\text{Sec}[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(b*f*(2*m+1)) +
  (a*A*m+b*B*(m+1))/(a^2*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && EqQ[m+n+1,0] && LeQ[m,-1]
```

2: $\int (a+b\,Sec[e+f\,x])^m\,(d\,Sec[e+f\,x])^n\,(A+B\,Sec[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,m+n+1==0\,\wedge\,m\nleq -1$

Derivation: Singly degenerate secant recurrence 1c with $m\to -n-2,\ p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,m+n+1==0\,\wedge\,m\nleq -1$, then

$$\int (a+b\,Sec[e+f\,x])^m\,(d\,Sec[e+f\,x])^n\,(A+B\,Sec[e+f\,x])\,dx\longrightarrow$$
$$-\frac{A\,Tan[e+f\,x]\,(a+b\,Sec[e+f\,x])^m\,(d\,Sec[e+f\,x])^n}{f\,n}-\frac{(a\,A\,m-b\,B\,n)}{b\,d\,n}\int (a+b\,Sec[e+f\,x])^m\,(d\,Sec[e+f\,x])^{n+1}\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*n) -
  (a*A*m-b*B*n)/(b*d*n)*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f,A,B,m,n},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && EqQ[m+n+1,0] && Not[LeQ[m,-1]]
```

2. $\int (a+b\,Sec[e+f\,x])^m\,(d\,Sec[e+f\,x])^n\,(A+B\,Sec[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\bigwedge\,a^2-b^2==0\,\bigwedge\,m\geq \frac{1}{2}$

1. $\int \sqrt{a+b\,Sec[e+f\,x]}\,(d\,Sec[e+f\,x])^n\,(A+B\,Sec[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2==0$

1: $\int \sqrt{a+b\,Sec[e+f\,x]}\,(d\,Sec[e+f\,x])^n\,(A+B\,Sec[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,A\,b\,(2\,n+1)+2\,a\,B\,n==0$

Derivation: Singly degenerate secant recurrence 1a with $B\to -\frac{A\,b\,(3+2\,n)}{2\,a\,(1+n)},\ m\to \frac{1}{2},\ p\to 0$

Derivation: Singly degenerate secant recurrence 1b with $B\to -\frac{A\,b\,(3+2\,n)}{2\,a\,(1+n)},\ m\to \frac{1}{2},\ p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,A\,b\,(2\,n+1)+2\,a\,B\,n==0$, then

$$\int \sqrt{a+b\,Sec[e+f\,x]}\,(d\,Sec[e+f\,x])^n\,(A+B\,Sec[e+f\,x])\,dx\longrightarrow\frac{2\,b\,B\,Tan[e+f\,x]\,(d\,Sec[e+f\,x])^n}{f\,(2\,n+1)\,\sqrt{a+b\,Sec[e+f\,x]}}$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -2*b*B*Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*(2*n+1)*Sqrt[a+b*Csc[e+f*x]]) /;
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && EqQ[A*b*(2*n+1)+2*a*B*n,0]
```

2. $\int\sqrt{a+b\,Sec[e+f\,x]}\,(d\,Sec[e+f\,x])^n\,(A+B\,Sec[e+f\,x])\,dx$ when $A\,b-a\,B\neq0\,\wedge\,a^2-b^2==0\,\wedge\,A\,b\,(2\,n+1)+2\,a\,B\,n\neq0$

1: $\int\sqrt{a+b\,Sec[e+f\,x]}\,(d\,Sec[e+f\,x])^n\,(A+B\,Sec[e+f\,x])\,dx$ when $A\,b-a\,B\neq0\,\wedge\,a^2-b^2==0\,\wedge\,A\,b\,(2\,n+1)+2\,a\,B\,n\neq0\,\wedge\,n<0$

Derivation: Singly degenerate secant recurrence 1a with $m\to\frac{1}{2},\ p\to0$

Rule: If $A\,b-a\,B\neq0\,\wedge\,a^2-b^2==0\,\wedge\,A\,b\,(2\,n+1)+2\,a\,B\,n\neq0\,\wedge\,n<0$**, then**

$$\int\sqrt{a+b\,Sec[e+f\,x]}\,(d\,Sec[e+f\,x])^n\,(A+B\,Sec[e+f\,x])\,dx\longrightarrow$$
$$-\frac{A\,b^2\,Tan[e+f\,x]\,(d\,Sec[e+f\,x])^n}{a\,f\,n\,\sqrt{a+b\,Sec[e+f\,x]}}+\frac{(A\,b\,(2\,n+1)+2\,a\,B\,n)}{2\,a\,d\,n}\int\sqrt{a+b\,Sec[e+f\,x]}\,(d\,Sec[e+f\,x])^{n+1}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  A*b^2*Cot[e+f*x]*(d*Csc[e+f*x])^n/(a*f*n*Sqrt[a+b*Csc[e+f*x]]) +
  (A*b*(2*n+1)+2*a*B*n)/(2*a*d*n)*Int[Sqrt[a+b*Csc[e+f*x]]*(d*Csc[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && NeQ[A*b*(2*n+1)+2*a*B*n,0] && LtQ[n,0]
```

2: $\int\sqrt{a+b\,Sec[e+f\,x]}\,(d\,Sec[e+f\,x])^n\,(A+B\,Sec[e+f\,x])\,dx$ when $A\,b-a\,B\neq0\,\wedge\,a^2-b^2==0\,\wedge\,A\,b\,(2\,n+1)+2\,a\,B\,n\neq0\,\wedge\,n\not<0$

Derivation: Singly degenerate secant recurrence 1b with $m\to\frac{1}{2},\ p\to0$

Rule: If $A\,b-a\,B\neq0\,\wedge\,a^2-b^2==0\,\wedge\,A\,b\,(2\,n+1)+2\,a\,B\,n\neq0\,\wedge\,n\not<0$**, then**

$$\int\sqrt{a+b\,Sec[e+f\,x]}\,(d\,Sec[e+f\,x])^n\,(A+B\,Sec[e+f\,x])\,dx\longrightarrow$$
$$\frac{2\,b\,B\,Tan[e+f\,x]\,(d\,Sec[e+f\,x])^n}{f\,(2\,n+1)\,\sqrt{a+b\,Sec[e+f\,x]}}+\frac{A\,b\,(2\,n+1)+2\,a\,B\,n}{b\,(2\,n+1)}\int\sqrt{a+b\,Sec[e+f\,x]}\,(d\,Sec[e+f\,x])^n\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -2*b*B*Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*(2*n+1)*Sqrt[a+b*Csc[e+f*x]]) +
  (A*b*(2*n+1)+2*a*B*n)/(b*(2*n+1))*Int[Sqrt[a+b*Csc[e+f*x]]*(d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && NeQ[A*b*(2*n+1)+2*a*B*n,0] && Not[LtQ[n,0]]
```

2. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\bigwedge\,a^2-b^2==0\,\bigwedge\,m>\frac{1}{2}$

1: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\bigwedge\,a^2-b^2==0\,\bigwedge\,m>\frac{1}{2}\,\bigwedge\,n<-1$

Derivation: Singly degenerate secant recurrence 1a with $p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\bigwedge\,a^2-b^2==0\,\bigwedge\,m>\frac{1}{2}\,\bigwedge\,n<-1$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\,\longrightarrow$$

$$-\frac{a\,A\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-1}\,(d\,\text{Sec}[e+f\,x])^n}{f\,n}-$$

$$\frac{b}{a\,d\,n}\int (a+b\,\text{Sec}[e+f\,x])^{m-1}\,(d\,\text{Sec}[e+f\,x])^{n+1}\,(a\,A\,(m-n-1)-b\,B\,n-(a\,B\,n+A\,b\,(m+n))\,\text{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  a*A*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^n/(f*n) -
  b/(a*d*n)*Int[(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^(n+1)*Simp[a*A*(m-n-1)-b*B*n-(a*B*n+A*b*(m+n))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && GtQ[m,1/2] && LtQ[n,-1]
```

2: $\int (a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Sec}[e + f x])^n (A + B \operatorname{Sec}[e + f x]) \, dx$ when $A b - a B \neq 0 \bigwedge a^2 - b^2 == 0 \bigwedge m > \frac{1}{2} \bigwedge n \nless -1$

Derivation: Singly degenerate secant recurrence 1b with $p \to 0$

- **Rule: If $A b - a B \neq 0 \bigwedge a^2 - b^2 == 0 \bigwedge m > \frac{1}{2} \bigwedge n \nless -1$, then**

$$\int (a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Sec}[e + f x])^n (A + B \operatorname{Sec}[e + f x]) \, dx \longrightarrow$$

$$\frac{b B \operatorname{Tan}[e + f x] (a + b \operatorname{Sec}[e + f x])^{m-1} (d \operatorname{Sec}[e + f x])^n}{f (m + n)} +$$

$$\frac{1}{d (m + n)} \int (a + b \operatorname{Sec}[e + f x])^{m-1} (d \operatorname{Sec}[e + f x])^n (a A d (m + n) + B (b d n) + (A b d (m + n) + a B d (2 m + n - 1)) \operatorname{Sec}[e + f x]) \, dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -b*B*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^n/(f*(m+n)) +
  1/(d*(m+n))*Int[(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^n*Simp[a*A*d*(m+n)+B*(b*d*n)+(A*b*d*(m+n)+a*B*d*(2*m+n-1))*Csc[e+f*x],x],x]
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && GtQ[m,1/2] && Not[LtQ[n,-1]]
```

3. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\bigwedge\,a^2-b^2==0\,\bigwedge\,m<-\frac{1}{2}$

1: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\bigwedge\,a^2-b^2==0\,\bigwedge\,m<-\frac{1}{2}\,\bigwedge\,n>0$

Derivation: Singly degenerate secant recurrence 2a with $p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\bigwedge\,a^2-b^2==0\,\bigwedge\,m<-\frac{1}{2}\,\bigwedge\,n>0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\,\longrightarrow$$

$$-\frac{d\,(A\,b-a\,B)\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^{n-1}}{a\,f\,(2\,m+1)}-$$

$$\frac{1}{a\,b\,(2\,m+1)}\int (a+b\,\text{Sec}[e+f\,x])^{m+1}\,(d\,\text{Sec}[e+f\,x])^{n-1}\,(A\,(a\,d\,(n-1))-B\,(b\,d\,(n-1))-d\,(a\,B\,(m-n+1)+A\,b\,(m+n))\,\text{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  d*(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-1)/(a*f*(2*m+1)) -
  1/(a*b*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)*
    Simp[A*(a*d*(n-1))-B*(b*d*(n-1))-d*(a*B*(m-n+1)+A*b*(m+n))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && LtQ[m,-1/2] && GtQ[n,0]
```

2: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\ \bigwedge\ a^2-b^2==0\ \bigwedge\ m<-\frac{1}{2}\ \bigwedge\ n\not>0$

Derivation: Singly degenerate secant recurrence 2b with $p\to 0$

Rule: If $A\,b-a\,B\neq 0\ \bigwedge\ a^2-b^2==0\ \bigwedge\ m<-\frac{1}{2}\ \bigwedge\ n\not>0$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])\,dx\longrightarrow$$
$$\frac{(A\,b-a\,B)\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n}{b\,f\,(2\,m+1)}-$$
$$\frac{1}{a^2\,(2\,m+1)}\int (a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^n\,(b\,B\,n-a\,A\,(2\,m+n+1)+(A\,b-a\,B)\,(m+n+1)\,\mathrm{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(b*f*(2*m+1)) -
  1/(a^2*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n*
    Simp[b*B*n-a*A*(2*m+n+1)+(A*b-a*B)*(m+n+1)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && LtQ[m,-1/2] && Not[GtQ[n,0]]
```

4: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\ \wedge\ a^2-b^2==0\ \wedge\ n>1$

Derivation: Singly degenerate secant recurrence 2c with $p\to 0$

Rule: If $A\,b-a\,B\neq 0\ \wedge\ a^2-b^2==0\ \wedge\ n>1$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])\,dx\longrightarrow$$
$$\frac{B\,d\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^{n-1}}{f\,(m+n)}+$$
$$\frac{d}{b\,(m+n)}\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^{n-1}\,(b\,B\,(n-1)+(A\,b\,(m+n)+a\,B\,m)\,\mathrm{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -B*d*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-1)/(f*(m+n)) +
  d/(b*(m+n))*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-1)*Simp[b*B*(n-1)+(A*b*(m+n)+a*B*m)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B,m},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && GtQ[n,1]
```

5: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\land\,a^2-b^2==0\,\land\,n<0$

Derivation: Singly degenerate secant recurrence 1c with $p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\land\,a^2-b^2==0\,\land\,n<0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\,\longrightarrow$$
$$-\frac{A\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n}{f\,n}-\frac{1}{b\,d\,n}\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^{n+1}\,(a\,A\,m-b\,B\,n-A\,b\,(m+n+1)\,\text{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*n) -
  1/(b*d*n)*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n+1)*Simp[a*A*m-b*B*n-A*b*(m+n+1)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B,m},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && LtQ[n,0]
```

6: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\land\,a^2-b^2==0$

Derivation: Algebraic expansion

Baisi: $A+B\,z==\frac{A\,b-a\,B}{b}+\frac{B\,(a+b\,z)}{b}$

Rule: If $A\,b-a\,B\neq 0\,\land\,a^2-b^2==0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\,\longrightarrow$$
$$\frac{A\,b-a\,B}{b}\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx+\frac{B}{b}\int (a+b\,\text{Sec}[e+f\,x])^{m+1}\,(d\,\text{Sec}[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  (A*b-a*B)/b*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n,x] +
  B/b*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,d,e,f,A,B,m},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0]
```

5. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0$

1. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m>1$

1. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m>1\,\wedge\,n\leq -1$

1: $\int (a+b\,\text{Sec}[e+f\,x])^2\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,n\leq -1$

Derivation: Nondegenerate secant recurrence 1a with $A\to a\,A,\ B\to A\,b+a\,B,\ C\to b\,B,\ m\to m-1,\ p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,n\leq -1$**, then**

$$\int (a+b\,\text{Sec}[e+f\,x])^2\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\longrightarrow$$

$$-\frac{a^2\,A\,\text{Sin}[e+f\,x]\,(d\,\text{Sec}[e+f\,x])^{n+1}}{d\,f\,n}+$$

$$\frac{1}{d\,n}\int (d\,\text{Sec}[e+f\,x])^{n+1}\left(a\,(2\,A\,b+a\,B)\,n+\left(2\,a\,b\,B\,n+A\left(b^2\,n+a^2\,(n+1)\right)\right)\text{Sec}[e+f\,x]+b^2\,B\,n\,\text{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^2*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  a^2*A*Cos[e+f*x]*(d*Csc[e+f*x])^(n+1)/(d*f*n) +
  1/(d*n)*Int[(d*Csc[e+f*x])^(n+1)*(a*(2*A*b+a*B)*n+(2*a*b*B*n+A*(b^2*n+a^2*(n+1)))*Csc[e+f*x]+b^2*B*n*Csc[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LeQ[n,-1]
```

2: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m>1\,\wedge\,n\leq -1$

Derivation: Nondegenerate secant recurrence 1a with $A\to a\,A,\ B\to A\,b+a\,B,\ C\to b\,B,\ m\to m-1,\ p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m>1\,\wedge\,n\leq -1$**, then**

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\ \longrightarrow$$
$$-\frac{a\,A\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-1}\,(d\,\text{Sec}[e+f\,x])^n}{f\,n}+$$
$$\frac{1}{d\,n}\int (a+b\,\text{Sec}[e+f\,x])^{m-2}\,(d\,\text{Sec}[e+f\,x])^{n+1}\cdot$$
$$\left(a\,(a\,B\,n-A\,b\,(m-n-1))+\left(2\,a\,b\,B\,n+A\left(b^2\,n+a^2\,(1+n)\right)\right)\text{Sec}[e+f\,x]+b\,(b\,B\,n+a\,A\,(m+n))\,\text{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  a*A*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^n/(f*n) +
  1/(d*n)*Int[(a+b*Csc[e+f*x])^(m-2)*(d*Csc[e+f*x])^(n+1)*
    Simp[a*(a*B*n-A*b*(m-n-1))+(2*a*b*B*n+A*(b^2*n+a^2*(1+n)))*Csc[e+f*x]+b*(b*B*n+a*A*(m+n))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && GtQ[m,1] && LeQ[n,-1]
```

2: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m>1\,\wedge\,n\nleq -1$

Derivation: Nondegenerate secant recurrence 1b with $A\to a\,A,\ B\to A\,b+a\,B,\ C\to b\,B,\ m\to m-1,\ p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m>1\,\wedge\,n\nleq -1$**, then**

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\longrightarrow$$

$$\frac{b\,B\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-1}\,(d\,\text{Sec}[e+f\,x])^n}{f\,(m+n)}+$$

$$\frac{1}{m+n}\int (a+b\,\text{Sec}[e+f\,x])^{m-2}\,(d\,\text{Sec}[e+f\,x])^n\,\cdot$$

$$\left(a^2\,A\,(m+n)+a\,b\,B\,n+\left(a\,(2\,A\,b+a\,B)\,(m+n)+b^2\,B\,(m+n-1)\right)\,\text{Sec}[e+f\,x]+b\,(A\,b\,(m+n)+a\,B\,(2\,m+n-1))\,\text{Sec}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -b*B*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^n/(f*(m+n)) +
  1/(m+n)*Int[(a+b*Csc[e+f*x])^(m-2)*(d*Csc[e+f*x])^n*
    Simp[a^2*A*(m+n)+a*b*B*n+(a*(2*A*b+a*B)*(m+n)+b^2*B*(m+n-1))*Csc[e+f*x]+b*(A*b*(m+n)+a*B*(2*m+n-1))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && GtQ[m,1] && Not[IGtQ[n,1] && Not[IntegerQ[m]]]
```

2. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1$

1. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,n>0$

1: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,0<n<1$

Derivation: Nondegenerate secant recurrence 1a with $C\to 0,\ p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,0<n<1$**, then**

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\longrightarrow$$

$$\frac{d\,(A\,b-a\,B)\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m+1}\,(d\,\text{Sec}[e+f\,x])^{n-1}}{f\,(m+1)\,\left(a^2-b^2\right)}+$$

$$\frac{1}{(m+1)\,\left(a^2-b^2\right)}\int (a+b\,\text{Sec}[e+f\,x])^{m+1}\,(d\,\text{Sec}[e+f\,x])^{n-1}\,\cdot$$

$$\left(d\,(n-1)\,(A\,b-a\,B)+d\,(a\,A-b\,B)\,(m+1)\,\text{Sec}[e+f\,x]-d\,(A\,b-a\,B)\,(m+n+1)\,\text{Sec}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -d*(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)/(f*(m+1)*(a^2-b^2)) +
  1/((m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)*
    Simp[d*(n-1)*(A*b-a*B)+d*(a*A-b*B)*(m+1)*Csc[e+f*x]-d*(A*b-a*B)*(m+n+1)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[m,-1] && LtQ[0,n,1]
```

2. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,n>1$

1: $\int \text{Sec}[e+f\,x]^3\,(a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1$

Derivation: Nondegenerate secant recurrence 1a with $A\to a\,A,\ B\to A\,b+a\,B,\ C\to b\,B,\ m\to m-1,\ p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1$**, then**

$$\int \text{Sec}[e+f\,x]^3\,(a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{a^2\,(A\,b-a\,B)\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m+1}}{b^2\,f\,(m+1)\,\left(a^2-b^2\right)}+\frac{1}{b^2\,(m+1)\,\left(a^2-b^2\right)}\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m+1}\cdot$$
$$\left(a\,b\,(A\,b-a\,B)\,(m+1)-(A\,b-a\,B)\,\left(a^2+b^2\,(m+1)\right)\,\text{Sec}[e+f\,x]+b\,B\,(m+1)\,\left(a^2-b^2\right)\,\text{Sec}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^3*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -a^2*(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b^2*f*(m+1)*(a^2-b^2)) +
  1/(b^2*(m+1)*(a^2-b^2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*
    Simp[a*b*(A*b-a*B)*(m+1)-(A*b-a*B)*(a^2+b^2*(m+1))*Csc[e+f*x]+b*B*(m+1)*(a^2-b^2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[m,-1]
```

2: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,n>1$

Derivation: Nondegenerate secant recurrence 1a with $A\to a\,A,\ B\to A\,b+a\,B,\ C\to b\,B,\ m\to m-1,\ p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,n>1$**, then**

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])\,dx \longrightarrow$$

$$-\frac{a\,d^2\,(A\,b-a\,B)\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n-2}}{b\,f\,(m+1)\,\left(a^2-b^2\right)}-\frac{d}{b\,(m+1)\,\left(a^2-b^2\right)}\int (a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n-2}\cdot$$
$$\left(a\,d\,(A\,b-a\,B)\,(n-2)+b\,d\,(A\,b-a\,B)\,(m+1)\,\mathrm{Sec}[e+f\,x]-\left(a\,A\,b\,d\,(m+n)-d\,B\,\left(a^2\,(n-1)+b^2\,(m+1)\right)\right)\mathrm{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  a*d^2*(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-2)/(b*f*(m+1)*(a^2-b^2)) -
  d/(b*(m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-2)*
    Simp[a*d*(A*b-a*B)*(n-2)+b*d*(A*b-a*B)*(m+1)*Csc[e+f*x]-(a*A*b*d*(m+n)-d*B*(a^2*(n-1)+b^2*(m+1)))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[m,-1] && GtQ[n,1]
```

2: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,n\not>0$

Derivation: Nondegenerate secant recurrence 1c with $C\to 0,\ p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,n\not>0$**, then**

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])\,dx \longrightarrow$$

$$-\frac{b\,(A\,b-a\,B)\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n}}{a\,f\,(m+1)\,\left(a^2-b^2\right)}+\frac{1}{a\,(m+1)\,\left(a^2-b^2\right)}\int (a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n}\cdot$$
$$\left(A\,\left(a^2\,(m+1)-b^2\,(m+n+1)\right)+a\,b\,B\,n-a\,(A\,b-a\,B)\,(m+1)\,\mathrm{Sec}[e+f\,x]+b\,(A\,b-a\,B)\,(m+n+2)\,\mathrm{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  b*(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n/(a*f*(m+1)*(a^2-b^2)) +
  1/(a*(m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n*
    Simp[A*(a^2*(m+1)-b^2*(m+n+1))+a*b*B*n-a*(A*b-a*B)*(m+1)*Csc[e+f*x]+b*(A*b-a*B)*(m+n+2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[m,-1] && Not[ILtQ[m+1/2,0] && ILtQ[n,0]]
```

3. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,0<m<1$

1: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,0<m<1\,\wedge\,n>0$

Derivation: Nondegenerate secant recurrence 1b with $A\to A\,c,\ B\to B\,c+A\,d,\ C\to B\,d,\ n\to n-1,\ p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,0<m<1\,\wedge\,n>0$**, then**

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{B\,d\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^{n-1}}{f\,(m+n)}+$$

$$\frac{d}{m+n}\int (a+b\,\text{Sec}[e+f\,x])^{m-1}\,(d\,\text{Sec}[e+f\,x])^{n-1}\,\cdot$$

$$\left(a\,B\,(n-1)+(b\,B\,(m+n-1)+a\,A\,(m+n))\,\text{Sec}[e+f\,x]+(a\,B\,m+A\,b\,(m+n))\,\text{Sec}[e+f\,x]^2\right)dx$$

–

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -B*d*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-1)/(f*(m+n)) +
  d/(m+n)*Int[(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^(n-1)*
    Simp[a*B*(n-1)+(b*B*(m+n-1)+a*A*(m+n))*Csc[e+f*x]+(a*B*m+A*b*(m+n))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[0,m,1] && GtQ[n,0]
```

2: $\int (a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Sec}[e + f x])^n (A + B \operatorname{Sec}[e + f x]) \, dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge 0 < m < 1 \wedge n \leq -1$

Derivation: Nondegenerate secant recurrence 1a with $C \to 0$, $p \to 0$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge 0 < m < 1 \wedge n \leq -1$, then

$$\int (a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Sec}[e + f x])^n (A + B \operatorname{Sec}[e + f x]) \, dx \longrightarrow$$

$$-\frac{A \operatorname{Tan}[e + f x] (a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Sec}[e + f x])^n}{f n} - \frac{1}{d n} \int (a + b \operatorname{Sec}[e + f x])^{m-1} (d \operatorname{Sec}[e + f x])^{n+1} \cdot$$
$$\left(A b m - a B n - (b B n + a A (n + 1)) \operatorname{Sec}[e + f x] - A b (m + n + 1) \operatorname{Sec}[e + f x]^2\right) dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*n) -
  1/(d*n)*Int[(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^(n+1)*
    Simp[A*b*m-a*B*n-(b*B*n+a*A*(n+1))*Csc[e+f*x]-A*b*(m+n+1)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[0,m,1] && LeQ[n,-1]
```

4: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\land\,a^2-b^2\neq 0\,\land\,n>1\,\land\,m+n\neq 0$

Derivation: Nondegenerate secant recurrence 1b with $A\to a\,A,\ B\to A\,b+a\,B,\ C\to b\,B,\ m\to m-1,\ p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\land\,a^2-b^2\neq 0\,\land\,n>1\,\land\,m+n\neq 0$**, then**

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\longrightarrow$$

$$\frac{B\,d^2\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m+1}\,(d\,\text{Sec}[e+f\,x])^{n-2}}{b\,f\,(m+n)}+$$

$$\frac{d^2}{b\,(m+n)}\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^{n-2}\,\left(a\,B\,(n-2)+B\,b\,(m+n-1)\,\text{Sec}[e+f\,x]+(A\,b\,(m+n)-a\,B\,(n-1))\,\text{Sec}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -B*d^2*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-2)/(b*f*(m+n)) +
  d^2/(b*(m+n))*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-2)*
     Simp[a*B*(n-2)+B*b*(m+n-1)*Csc[e+f*x]+(A*b*(m+n)-a*B*(n-1))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,m},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && GtQ[n,1] && NeQ[m+n,0] && Not[IGtQ[m,1]]
```

5: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,n\leq -1$

Derivation: Nondegenerate secant recurrence 1c with $C\to 0,\ p\to 0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,n\leq -1$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\;\longrightarrow$$
$$-\frac{A\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m+1}\,(d\,\text{Sec}[e+f\,x])^n}{a\,f\,n}+$$
$$\frac{1}{a\,d\,n}\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^{n+1}\left(a\,B\,n-A\,b\,(m+n+1)+A\,a\,(n+1)\,\text{Sec}[e+f\,x]+A\,b\,(m+n+2)\,\text{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n/(a*f*n) +
  1/(a*d*n)*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n+1)*
    Simp[a*B*n-A*b*(m+n+1)+A*a*(n+1)*Csc[e+f*x]+A*b*(m+n+2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,m},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LeQ[n,-1]
```

6: $\int \frac{A+B\,\text{Sec}[e+f\,x]}{\sqrt{d\,\text{Sec}[e+f\,x]}\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B\,z}{\sqrt{d\,z}\,\sqrt{a+b\,z}}==\frac{A\,\sqrt{a+b\,z}}{a\,\sqrt{d\,z}}-\frac{(A\,b-a\,B)\,\sqrt{d\,z}}{a\,d\,\sqrt{a+b\,z}}$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0$, then

$$\int \frac{A+B\,\text{Sec}[e+f\,x]}{\sqrt{d\,\text{Sec}[e+f\,x]}\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx\;\longrightarrow\;\frac{A}{a}\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{d\,\text{Sec}[e+f\,x]}}\,dx-\frac{A\,b-a\,B}{a\,d}\int \frac{\sqrt{d\,\text{Sec}[e+f\,x]}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$$

Program code:

```
Int[(A_+B_.*csc[e_.+f_.*x_])/(Sqrt[d_.*csc[e_.+f_.*x_]]*Sqrt[a_+b_.*csc[e_.+f_.*x_]]),x_Symbol] :=
  A/a*Int[Sqrt[a+b*Csc[e+f*x]]/Sqrt[d*Csc[e+f*x]],x] -
  (A*b-a*B)/(a*d)*Int[Sqrt[d*Csc[e+f*x]]/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0]
```

7: $\int \frac{\sqrt{d \operatorname{Sec}[e+f x]}\,(A+B \operatorname{Sec}[e+f x])}{\sqrt{a+b \operatorname{Sec}[e+f x]}}\,dx$ when $A b-a B\neq 0 \wedge a^2-b^2\neq 0$

Derivation: Algebraic expansion

Rule: If $A b-a B\neq 0 \wedge a^2-b^2\neq 0$, then

$$\int \frac{\sqrt{d \operatorname{Sec}[e+f x]}\,(A+B \operatorname{Sec}[e+f x])}{\sqrt{a+b \operatorname{Sec}[e+f x]}}\,dx \to A \int \frac{\sqrt{d \operatorname{Sec}[e+f x]}}{\sqrt{a+b \operatorname{Sec}[e+f x]}}\,dx+\frac{B}{d}\int \frac{(d \operatorname{Sec}[e+f x])^{3/2}}{\sqrt{a+b \operatorname{Sec}[e+f x]}}\,dx$$

Program code:

```
Int[Sqrt[d_.*csc[e_.+f_.*x_]]*(A_+B_.*csc[e_.+f_.*x_])/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  A*Int[Sqrt[d*Csc[e+f*x]]/Sqrt[a+b*Csc[e+f*x]],x] +
  B/d*Int[(d*Csc[e+f*x])^(3/2)/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0]
```

8: $\int \frac{\sqrt{a+b \operatorname{Sec}[e+f x]}\,(A+B \operatorname{Sec}[e+f x])}{\sqrt{d \operatorname{Sec}[e+f x]}}\,dx$ when $A b-a B\neq 0 \wedge a^2-b^2\neq 0$

Derivation: Algebraic expansion

- **Basis:** $\frac{A+B z}{\sqrt{d z}} == \frac{B\sqrt{d z}}{d}+\frac{A}{\sqrt{d z}}$

Rule: If $A b-a B\neq 0 \wedge a^2-b^2\neq 0$, then

$$\int \frac{\sqrt{a+b \operatorname{Sec}[e+f x]}\,(A+B \operatorname{Sec}[e+f x])}{\sqrt{d \operatorname{Sec}[e+f x]}}\,dx \to \frac{B}{d}\int \sqrt{a+b \operatorname{Sec}[e+f x]}\,\sqrt{d \operatorname{Sec}[e+f x]}\,dx+A \int \frac{\sqrt{a+b \operatorname{Sec}[e+f x]}}{\sqrt{d \operatorname{Sec}[e+f x]}}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(A_+B_.*csc[e_.+f_.*x_])/Sqrt[d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  B/d*Int[Sqrt[a+b*Csc[e+f*x]]*Sqrt[d*Csc[e+f*x]],x] +
  A*Int[Sqrt[a+b*Csc[e+f*x]]/Sqrt[d*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0]
```

9: $\int \frac{(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])}{a+b\,\mathrm{Sec}[e+f\,x]}\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B\,z}{a+b\,z}==\frac{A}{a}-\frac{(A\,b-a\,B)\,(d\,z)}{a\,d\,(a+b\,z)}$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0$**, then**

$$\int \frac{(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])}{a+b\,\mathrm{Sec}[e+f\,x]}\,dx \longrightarrow \frac{A}{a}\int (d\,\mathrm{Sec}[e+f\,x])^n\,dx-\frac{A\,b-a\,B}{a\,d}\int \frac{(d\,\mathrm{Sec}[e+f\,x])^{n+1}}{a+b\,\mathrm{Sec}[e+f\,x]}\,dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_])/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  A/a*Int[(d*Csc[e+f*x])^n,x] - (A*b-a*B)/(a*d)*Int[(d*Csc[e+f*x])^(n+1)/(a+b*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0]
```

X: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])\,dx$ when $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0$

Rule: If $A\,b-a\,B\neq 0\,\wedge\,a^2-b^2\neq 0$**, then**

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])\,dx \longrightarrow$$
$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_.*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  Unintegrable[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n*(A+B*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,A,B,m,n},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0]
```

Rules for integrands of the form $(a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])^p$

1. $\int(a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])^p\,dx$ when $b\,c+a\,d==0\wedge a^2-b^2==0$

x: $\int(a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])^p\,dx$ when $b\,c+a\,d==0\wedge a^2-b^2==0\wedge m\in\mathbb{Z}$

- **Derivation: Algebraic simplification**
- **Basis: If** $b\,c+a\,d==0\wedge a^2-b^2==0$, **then** $(a+b\,\mathrm{Sec}[z])\,(c+d\,\mathrm{Sec}[z])==-a\,c\,\mathrm{Tan}[z]^2$
- **Rule: If** $b\,c+a\,d==0\wedge a^2-b^2==0\wedge m\in\mathbb{Z}$, **then**

$$\int(a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])^p\,dx\longrightarrow(-a\,c)^m\int\mathrm{Tan}[e+f\,x]^{2m}\,(c+d\,\mathrm{Sec}[e+f\,x])^{n-m}\,(A+B\,\mathrm{Sec}[e+f\,x])^p\,dx$$

- **Program code:**

```
(* Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_.*(A_.+B_.*csc[e_.+f_.*x_])^p_.,x_Symbol] :=
  (-a*c)^m*Int[Cot[e+f*x]^(2*m)*(c+d*Csc[e+f*x])^(n-m)*(A+B*Csc[e+f*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,A,B,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[m] &&
  Not[IntegerQ[n] && (LtQ[m,0] && GtQ[n,0] || LtQ[0,n,m] || LtQ[m,n,0])] *)
```

1: $\int(a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])^p\,dx$ when $b\,c+a\,d==0\wedge a^2-b^2==0\wedge(m\mid n\mid p)\in\mathbb{Z}$

- **Derivation: Algebraic simplification**
- **Basis: If** $b\,c+a\,d==0\wedge a^2-b^2==0$, **then** $(a+b\,\mathrm{Sec}[z])\,(c+d\,\mathrm{Sec}[z])==-a\,c\,\mathrm{Tan}[z]^2$
- **Rule: If** $b\,c+a\,d==0\wedge a^2-b^2==0\wedge(m\mid n\mid p)\in\mathbb{Z}$, **then**

$$\int(a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x])^p\,dx\longrightarrow(-a\,c)^m\int\mathrm{Tan}[e+f\,x]^{2m}\,(c+d\,\mathrm{Sec}[e+f\,x])^{n-m}\,(A+B\,\mathrm{Sec}[e+f\,x])^p\,dx$$

$$\longrightarrow(-a\,c)^m\int\frac{\mathrm{Sin}[e+f\,x]^{2m}\,(d+c\,\mathrm{Cos}[e+f\,x])^{n-m}\,(B+A\,\mathrm{Cos}[e+f\,x])^p}{\mathrm{Cos}[e+f\,x]^{m+n+p}}\,dx$$

- **Program code:**

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_.*(A_.+B_.*csc[e_.+f_.*x_])^p_.,x_Symbol] :=
  (-a*c)^m*Int[Cos[e+f*x]^(2*m)*(d+c*Sin[e+f*x])^(n-m)*(B+A*Sin[e+f*x])^p/Sin[e+f*x]^(m+n+p),x] /;
FreeQ[{a,b,c,d,e,f,A,B,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegersQ[m,n,p]
```